Laborator 3: Sisteme de ecuatii diferentiale

In cadrul acestui laborator vom studia sisteme de ecuatii diferentiale utilizand MAPLE 8. Sunt prezenate procedurile necesare pentru obtinerea graficelelor corespunzatoare solutiilor in cazul unei probleme cu valori initiale (problema Cauchy), precum si reprezentarea campului de directii si a portretului fazic corespunzator.

Rezolvarea unui sistem de ecuatii diferentiale

Se considera sistemul liniar de doua ecuatii diferentiale:
x' = x+y
y' = x-y
Sistemul dat se introduce definind cele doua ecuatii prin intermediul procedurii **diff**

```
> restart;
> ec1:=diff(x(t),t)=x(t)+y(t);
```

$$ec1 := \frac{d}{dt}\,\mathrm{x}(t) = \mathrm{x}(t) + \mathrm{y}(t)$$

```
> ec2:=diff(y(t),t)=x(t)-y(t);
```

$$ec2 := \frac{d}{dt}\,\mathrm{y}(t) = \mathrm{x}(t) - \mathrm{y}(t)$$

```
> sist:=ec1,ec2;
```

$$sist := \frac{d}{dt}\,\mathrm{x}(t) = \mathrm{x}(t) + \mathrm{y}(t),\ \frac{d}{dt}\,\mathrm{y}(t) = \mathrm{x}(t) - \mathrm{y}(t)$$

Pentru obtinerea solutiei generale a sistemul se utilizeaza comanda **dsolve** cunoscuta din cadrul ecuatiilor diferentiale:

```
> with(DEtools): with(plots):
Warning, the name changecoords has been redefined
```

```
> dsolve({sist},{x(t),y(t)});
```

$$\{\mathrm{x}(t) = \_C1\,\mathbf{e}^{(\sqrt{2}\,t)} + \_C2\,\mathbf{e}^{(-\sqrt{2}\,t)},$$
$$\mathrm{y}(t) = \_C1\,\sqrt{2}\,\mathbf{e}^{(\sqrt{2}\,t)} - \_C2\,\sqrt{2}\,\mathbf{e}^{(-\sqrt{2}\,t)} - \_C1\,\mathbf{e}^{(\sqrt{2}\,t)} - \_C2\,\mathbf{e}^{(-\sqrt{2}\,t)}\}$$

Pentru reprezentarea grafica a solutiilor se definesc functiile solutii utilizand comenzile **rhs** si **unapply** si apoi se dau valori celor doua constante de integrare *_C1* si *_C2*

```
> sol:=dsolve({sist},{x(t),y(t)});
```

$$sol := \{\mathrm{x}(t) = \_C1\,\mathbf{e}^{(\sqrt{2}\,t)} + \_C2\,\mathbf{e}^{(-\sqrt{2}\,t)},$$
$$\mathrm{y}(t) = \_C1\,\sqrt{2}\,\mathbf{e}^{(\sqrt{2}\,t)} - \_C2\,\sqrt{2}\,\mathbf{e}^{(-\sqrt{2}\,t)} - \_C1\,\mathbf{e}^{(\sqrt{2}\,t)} - \_C2\,\mathbf{e}^{(-\sqrt{2}\,t)}\}$$

Atentie variabila *sol* este o lista, accesul la x(t) si y(t) se face prin **sol[1]**, respectiv **sol[2]**

```
> xx:=unapply(rhs(sol[1]),t,_C1,_C2);
```

$$xx := (t, \_C1, \_C2) \rightarrow \_C1\, \mathbf{e}^{(\sqrt{2}\, t)} + \_C2\, \mathbf{e}^{(-\sqrt{2}\, t)}$$

```
> yy:=unapply(rhs(sol[2]),t,_C1,_C2);
```

$$yy := (t, \_C1, \_C2) \rightarrow \_C1\, \sqrt{2}\, \mathbf{e}^{(\sqrt{2}\, t)} - \_C2\, \sqrt{2}\, \mathbf{e}^{(-\sqrt{2}\, t)} - \_C1\, \mathbf{e}^{(\sqrt{2}\, t)} - \_C2\, \mathbf{e}^{(-\sqrt{2}\, t)}$$

In variabilele **xx** si **yy** avem expresiile solutiilor sub forma unor functii ce depind de variabila independenta si cele doua constante de integrare. Daca dorim sa obtinem graficele solutiilor pentru constantele **_C1=1** si **_C2=1** se utilizeaza comanda **plot**

```
> plot(xx(t,1,1),t=-4..4);
```




```
> plot(yy(t,1,1),t=-4..4);
```



sau pot fi reprezentate ambele solutii in acelasi grafic:

> **plot([xx(t,1,1),yy(t,1,1)],t=-4..4,color=[red,blue]);**



Probleme Cauchy

Daca dorim obtinerea solutiilor corespunzatoare unei probleme cu valori initiale, de exemplu:

x' = x+y

y' = x-y

x(0) = 1

y(0) = 0

se definesc conditiile initiale ca in cazul ecuatiilor diferentiale:

> **cond_in:=x(0)=1, y(0)=0;**

$$cond\_in := x(0) = 1, y(0) = 0$$

iar pentru obtinerea solutiilor se utilizeaza, din nou, comanda **dsolve** precizandu-se conditiile initiale:

> **sol:=dsolve({sist,cond_in},{x(t),y(t)});**

$$sol := \{ x(t) = \left(\frac{1}{2} + \frac{\sqrt{2}}{4}\right) e^{(\sqrt{2}\, t)} + \left(\frac{1}{2} - \frac{\sqrt{2}}{4}\right) e^{(-\sqrt{2}\, t)}, y(t) = \left(\frac{1}{2} + \frac{\sqrt{2}}{4}\right) \sqrt{2}\, e^{(\sqrt{2}\, t)} - \left(\frac{1}{2} - \frac{\sqrt{2}}{4}\right) \sqrt{2}\, e^{(-\sqrt{2}\, t)} - \left(\frac{1}{2} + \frac{\sqrt{2}}{4}\right) e^{(\sqrt{2}\, t)} - \left(\frac{1}{2} - \frac{\sqrt{2}}{4}\right) e^{(-\sqrt{2}\, t)} \}$$

Pentru reprezentarea grafica a solutiilor avem doua alternative. Prima varianta este sa definim functiile solutii utilizand comenzile **rhs** si **unapply** (ca in cazul solutiei generale):

> **xx:=unapply(rhs(sol[1]),t);**

$$xx := t \to \left(\frac{1}{2} + \frac{\sqrt{2}}{4}\right) e^{(\sqrt{2}\, t)} + \left(\frac{1}{2} - \frac{\sqrt{2}}{4}\right) e^{(-\sqrt{2}\, t)}$$

> **yy:=unapply(rhs(sol[2]),t);**

$$yy := t \to \left(\frac{1}{2} + \frac{\sqrt{2}}{4}\right) \sqrt{2}\, e^{(\sqrt{2}\, t)} - \left(\frac{1}{2} - \frac{\sqrt{2}}{4}\right) \sqrt{2}\, e^{(-\sqrt{2}\, t)} - \left(\frac{1}{2} + \frac{\sqrt{2}}{4}\right) e^{(\sqrt{2}\, t)} - \left(\frac{1}{2} - \frac{\sqrt{2}}{4}\right) e^{(-\sqrt{2}\, t)}$$

si apoi utilizam comanda **plot**

```
> plot(xx(t),t=-5..5);
```




```
> plot(yy(t),t=-5..5);
```



sau pot fi reprezentate ambele solutii in acelasi grafic:

```
> plot([xx(t),yy(t)],t=-5..5,color=[red,blue]);
```



A doua varianta de reprezentare grafica a solutiilor este utlizarea comenzii **DEplot** cu optiunea **scene**

```
> DEplot([sist],[x,y],t=5..5,[[cond_in]],linecolor=blue,
scene=[t,x(t)]);
```




```
> DEplot([sist],[x,y],t=-5..5,[[cond_in]], linecolor=red,
scene=[t,y(t)]);
```



In cazul in care dorim reprezentarea grafica a ambelor solutii in acelasi grafic utilizand comanda **DEplot** vom utiliza intructiunile:

```
> xx1:=DEplot([sist],[x,y],t=-5..5,[[cond_in]], linecolor=blue,
scene=[t,x(t)]):
> yy1:=DEplot([sist],[x,y],t=-5..5,[[cond_in]],linecolor=red,
scene=[t,y(t)]):
> display([xx1,yy1]);
```



Campul de directii. Portret Fazic

Pentru reprezentarea campului de directii utilizam instructiunea:

```
> DEplot([sist],[x(t),y(t)],t=-4..4,x=-10..10,y=-10..10,
arrows=medium);
```



iar pentru reprezentarea portretului fazic (campul de directii si cateva orbite) vom utiliza aceeasi instructiune cu diferenta ca sunt precizate anumite conditii initiale reprezentand curbele ce trec prin punctele precizate:

```
> DEplot([sist],[x(t),y(t)],t=-4..4,x=-10..10,y=-10..10,
[[x(0)=1,y(0)=0], [x(0)=1,y(0)=5]], arrows=medium, linecolor=blue);
```



In acest grafic avem reprezentate doua orbite, daca dorim reprezentarea mai multor orbite trebuiesc precizate mai multe conditii initiale:

```
> cond_in2:=[x(0)=0,y(0)=i]$i=1..5, [x(0)=0,y(0)=-i]$i=1..5,
[x(0)=i,y(0)=0]$i=1..5, [x(0)=-i,y(0)=0]$i=1..5;
```

$cond\_in2 := [x(0)=0, y(0)=1], [x(0)=0, y(0)=2], [x(0)=0, y(0)=3],$
$[x(0)=0, y(0)=4], [x(0)=0, y(0)=5], [x(0)=0, y(0)=-1],$
$[x(0)=0, y(0)=-2], [x(0)=0, y(0)=-3], [x(0)=0, y(0)=-4],$
$[x(0)=0, y(0)=-5], [x(0)=1, y(0)=0], [x(0)=2, y(0)=0],$
$[x(0)=3, y(0)=0], [x(0)=4, y(0)=0], [x(0)=5, y(0)=0],$
$[x(0)=-1, y(0)=0], [x(0)=-2, y(0)=0], [x(0)=-3, y(0)=0],$
$[x(0)=-4, y(0)=0], [x(0)=-5, y(0)=0]$

```
> DEplot([sist],[x(t),y(t)],t=-5..5,x=-10..10,y=-10..10,[cond_in2],
arrows=medium, linecolor=blue,stepsize=0.1);
```