Laborator 1: Introducere in Maple

Calcule in Maple

Maple poate face calcule folosind folosind operatorii:
+ - adunare
- - scadere
* - inmultire
/ - impartire
^ - ridicarea la putere
! - factorial
sqrt( ) - radical
exp( ) - functia exponentiala
ln - functia logaritm
sin, cos, tan, cot - functiile trigonometrice sinus, cosinus, tangenta, cotangenta

```
> 1+2-3;
```

$$0$$

```
> 2*3/7+3^2;
```

$$\frac{69}{7}$$

```
> sqrt(100);
```

$$10$$

```
> 3^2;
```

$$9$$

```
> sqrt(5);
```

$$\sqrt{5}$$

```
> sqrt(5.0);
```

$$2.236067977$$

Cand este introdus un numar intreg in expresia radicalului, MAPLE executa un *calcul simbolic*, daca este introdus un numar zecimal, MAPLE executa un *calcul numeric* cu o precizie de 10 zecimale. Functia **evalf** returneaza valoarea numerica a expresiei precizate.

```
> evalf(sqrt(5));
```

$$2.236067977$$

```
> (1/5)^3;
```

$$\frac{1}{125}$$

```
> (0.2)^3;
```

$$0.008$$

```
> evalf((1/5)^3);
```

$$0.008000000000$$

Variabile

Se pot atribui valori unor variabile folosind comanda ":="

```
> x:=1;y:=2;
```

$$1$$
$$2$$

```
> (x^2+y^2)/(2*x*y);
```

$$\frac{5}{4}$$

```
> evalf(%);
```

$$1.250000000$$

Atunci cand dorim evaluarea numerica a expresiei precedente putem folosi comanda **evalf(%)**

Pot fi definite si cu litere grecesti:

```
> alpha, beta, gamma, Alpha, Beta, Gamma;
```

$$\alpha, \beta, \gamma, A, B, \Gamma$$

Observatie: Expresia **Pi** are atribuita valoarea numerica a acestui numar, pe cand expresia **pi** returneaza litera respectiva

```
> pi; evalf(pi);
```

$$\pi$$

$$\pi$$

```
> Pi; evalf(Pi);
```

$$\pi$$

$$3.141592654$$

```
> alpha:=3*Pi/4;
```

$$\frac{3}{4}\pi$$

```
> sin(alpha);cos(alpha);tan(alpha);cot(alpha);
```

$$\frac{1}{2}\sqrt{2}$$

$$-\frac{1}{2}\sqrt{2}$$

$$-1$$

$$-1$$

```
> ln(alpha),exp(alpha);
```

$$\ln\left(\frac{3}{4}\pi\right), e^{\frac{3}{4}\pi}$$

```
> evalf(%);
```

$$0.8570478133, 10.55072407$$

Functii. Reprezentari grafice

```
> restart;
```

Comanda **restart** elibereaza memoria de valorile utilizate.

Functiile pot fi definite ca operatori, apoi putand fi utilizate pentru diverse calcule sau expresii

```
> f:=x->sin(x)/x;
```

$$f := x \to \frac{\sin(x)}{x}$$

```
> f(3*Pi/2),f(1.5);
```

$$-\frac{2}{3\pi}, 0.6649966577$$

```
> f(a+b);
```

$$\frac{\sin(a+b)}{a+b}$$

Pentru reprezentarea grafica trebuie incarcat pachetul **plots** utilizand comanda **with**

```
> with(plots):
> plot(f(x),x=-1..1);
```



Pot fi reprezentate grafic mai multe functii in aceeasi fereastra precizandu-se lista de functii $[f_1(x), f_2(x), f_3(x)]$ si, daca dorim identificarea lor, lista de culori $[c_1, c_2, c_3]$

```
> plot([f(x),f(2*x),sin(x)],x=0..2*Pi,color=[red,blue,green]);
```



Daca lista de functii este mai mare se poate utiliza operatorul **$** in forma (**expr$i=m..n**) pentru generarea acesteia. Acesta returneza sirul obtinut prin inlocuirea lui **i** cuprins intre **m** si **n** in expresia **expr**. De exemplu, pentru sirul de functii $f_n(x) = \dfrac{x}{(1+x^2)^n}$ daca dorim reprezentarea functiilor $f_1(x)$, ..., $f_{10}(x)$ generam lista de functii

```
> f:=(x,n)->x/(1+x^2)^n;
```

$$f := (x, n) \to \frac{x}{(1+x^2)^n}$$

```
> list_f:=f(x,i)$i=1..10;
```

$$list\_f := \frac{x}{x^2+1}, \frac{x}{(x^2+1)^2}, \frac{x}{(x^2+1)^3}, \frac{x}{(x^2+1)^4}, \frac{x}{(x^2+1)^5}, \frac{x}{(x^2+1)^6}, \frac{x}{(x^2+1)^7}, \frac{x}{(x^2+1)^8}, \frac{x}{(x^2+1)^9}, \frac{x}{(x^2+1)^{10}}$$

```
> plot([list_f],x=-3..3);
```



In cazul in care functia are puncte de discontinuitate este utilizata optiunea **discont = true**

```
> plot(tan(x), x = -2*Pi..2*Pi, y = -4..4, discont = true);
```



Daca o expresia unei curbe este data in forma parametrica, de exemplu:

$$x(t) = \sin(t)$$

$$y(t) = \cos(t), t = 0 .. \frac{3}{2}\pi$$

se va utiliza comanda **plot** in forma:

```
> plot([sin(t),cos(t),t=0..3/2*Pi]);
```



Argumentul functiei plot, in acest caz, este o lista de 3 componente: **[x(t),y(t),t=a..b]** prima variabila reprezinta coordonata x, a doua coordonata y, a treia intervalul parcurs de parametru. Pentru a reprezenta in acelasi grafic mai multe curbe date in forma parametrica argumentul functiei plot va fi o lista de lista de curbe, de exemplu pentru curbele:

(C1): $x(t) = \sin(t)$ $y(t) = \cos(t)$, $t = 0..\pi$ (C2): $x(t) = 2\sin(t)$, $t = 0..2\pi$ $y(t) = \cos(t)$, $t = 0..2\pi$

comanda plot va fi de forma:

```
> plot([[sin(t),cos(t),t=0..3/2*Pi],[2*sin(t),cos(t),t=0..2*Pi]],
color=[red,blue],scaling=constrained);
```



In cazul in care avem o curba data in forma implicita se va utiliza comanda **implicitplot**: De exemplu, ecuatia cercului centrat in (0,0) si de raza r in forma implicita este de forma $x^2 + y^2 = r^2$. In cazul cercului de raza 1 vom utiliza comanda:

```
> implicitplot(x^2+y^2=1,x=-1..1,y=-1..1);
```



Daca dorim sa vizualizam dependenta unei functii fata de un parametru este utilizata comanda **animate** (se da clic dreapta pe imagine se selecteaza *Animation* si apoi *Play*)

```
> animate(sin(x*t),x=-4*Pi..4*Pi,t=0..2,color=red);
```



Daca dorim o precizie mai mare specificam numarul de puncte si numarul de frame-uri

```
> animate(sin(x*t),x=-4*Pi..4*Pi,t=0..2,color=red,numpoints=100,
frames=100);
```



Functiile de mai multe variabile pot fi definite in felul urmator:

```
> g:=(x,y)->sin(x)*exp(-y);
```

$$g := (x, y) \rightarrow \sin(x)\, e^{-y}$$

Reprezentarea grafica a functiilor de 2 variabile se face prin comanda **plot3d:**

```
> plot3d(g(x,y),x=0..Pi,y=0..3,axes=boxed);
```



Animatia graficelor 3D se face cu ajutorul comenzii **animate3d:**

```
> animate3d(g(t*x,y),x=0..Pi,y=0..3,t=0..2);
```



Limite, Derivate, Integrale

```
> restart:
```

Limitele de siruri si functii se calculeaza folosind comanda **limit**:

```
> limit(1/n,n=infinity);
```

$$0$$

```
> limit(sin(x)/x,x=0);
```

$$1$$

```
> limit(exp(x), x=infinity);
```

$$\infty$$

```
> limit(1/x, x=0, real);
```

$$undefined$$

Derivarea functiilor se face cu ajutorul comenzii **diff** sau utilizand operatorul de derivare **D**:

```
> f:=x->exp(x^2)+3;
```

$$f := x \rightarrow e^{x^2} + 3$$

Derivarea functiilor se face cu ajutorul comenzii **diff** sau utilizand operatorul de derivare **D**. Comanda **diff** executa derivarea expresiei respective in raport cu variabila precizata pe cand operatorul **D** ne returneaza derivata ca functie.

```
> diff(f(x),x);
```

$$2\,x\,e^{x^2}$$

derivata de ordinul 2

```
> diff(f(x),x,x);
```

$$2\,e^{x^2} + 4\,x^2\,e^{x^2}$$

derivata de ordinul n mai poate fi obtinuta utilizand optiunea **t$n**

```
> diff(f(x),x$2);
```

$$2\,e^{x^2} + 4\,x^2\,e^{x^2}$$

```
> diff(f(x),x$3);
```

$$12\,x\,e^{x^2} + 8\,x^3\,e^{x^2}$$

```
> D(f)(x);
```

$$2\,x\,e^{x^2}$$

Operatorul **D** este utilizat atunci cand avem nevoie de valorile derivatei functiei f in diferite puncte, de exemplu pentru calculul lui f '(0) sau f '(1) folosim:

```
> D(f)(0);D(f)(1);
```

$$0$$

$$2\,e$$

```
> (D@D)(f)(x);
```

$$2e^{x^2} + 4x^2 e^{x^2}$$

```
> (D@@2)(f)(x);
```

$$2e^{x^2} + 4x^2 e^{x^2}$$

```
> (D@D@D)(f)(x);
```

$$12xe^{x^2} + 8x^3 e^{x^2}$$

```
> (D@@3)(f)(x);
```

$$12xe^{x^2} + 8x^3 e^{x^2}$$

Calculul primitivelor si a integralelor definite se face cu ajutorul comenzii **int**:

```
> int(cos(x),x);
```

$$\sin(x)$$

daca se precizeaza limitele de integrare ***x=a..b*** se obtine valoarea integralei definite de la *a* la *b:*

```
> int(cos(x),x=0..Pi);
```

$$0$$

```
> int(1/x,x=1..infinity);
```

$$\infty$$

Nu intotdeauna MAPLE poate calcula integrala definita. De exemplu

```
> int( sin( sqrt(1 - x^3) ), x = 0..1 );
```

$$\int_0^1 \sin\left(\sqrt{-x^3+1}\right) dx$$

dar poate obtine o valoare aproximativa a acesteia utilizand metode numerice de aproximare:

```
> evalf( int( sin( sqrt(1 - x^3) ), x = 0..1 ) );
```

$$0.7315380065$$

Ecuatii si sisteme de ecuatii algebrice. Algebra liniara

Rezolvarea ecuatiilor si sistemelor de ecuatii algebrice se face prin comanda **solve**:

```
> restart:
> solve( x^2 + 3*x + 2=0 );
```

$$-1, -2$$

```
> solve( x^2 + x + 1=0 );
```

$$-\frac{1}{2} + \frac{1}{2} I\sqrt{3}, -\frac{1}{2} - \frac{1}{2} I\sqrt{3}$$

Comanda **fsolve** rezolva ecuatiile folosind metode numerice de aproximare a solutiilor:

```
> solve( x^3 + x = 27 );
```

$$\frac{1}{6}\left(2916+12\sqrt{59061}\right)^{1/3}-\frac{2}{\left(2916+12\sqrt{59061}\right)^{1/3}},-\frac{1}{12}\left(2916+12\sqrt{59061}\right)^{1/3}+\frac{1}{\left(2916+12\sqrt{59061}\right)^{1/3}}+\frac{1}{2}I\sqrt{3}\left(\frac{1}{6}\left(2916+12\sqrt{59061}\right)^{1/3}+\frac{2}{\left(2916+12\sqrt{59061}\right)^{1/3}}\right),-\frac{1}{12}\left(2916+12\sqrt{59061}\right)^{1/3}+\frac{1}{\left(2916+12\sqrt{59061}\right)^{1/3}}-\frac{1}{2}I\sqrt{3}\left(\frac{1}{6}\left(2916+12\sqrt{59061}\right)^{1/3}+\frac{2}{\left(2916+12\sqrt{59061}\right)^{1/3}}\right)$$

```
> fsolve( x^3 + x = 27 );
```

$$2.888941572$$

```
> solve( tan(x) - x = 2 );
```

$$RootOf(-\tan(\_Z)+\_Z+2)$$

```
> fsolve( tan(x) - x = 2 );
```

$$1.274392662$$

In cazul sistemelor de ecuatii se utilizeaza aceeasi comanda, ecuatiile punandu-se intre acolade

```
> solve({x+2*y=1,x-y=3},{x,y});
```

$$\left\{x=\frac{7}{3},y=-\frac{2}{3}\right\}$$

In cazul operatiilor cu vectori si matrici trebuie incarcat pachetul de algebra liniara **linalg**:

```
> with(linalg):
```

Vectorii pot fi definiti in modul urmator:

```
> v := vector( [1, -1] );
```

$$v:=\begin{bmatrix}1 & -1\end{bmatrix}$$

```
> v[1]; v[2];
```

$$1$$
$$-1$$

```
> print( v );
```

$$\begin{bmatrix}1 & -1\end{bmatrix}$$

```
> w := vector( [1, 1] );
```

$$w:=\begin{bmatrix}1 & 1\end{bmatrix}$$

```
> evalm(v+w);
```

$$\begin{bmatrix}2 & 0\end{bmatrix}$$

```
> dotprod( v, w );  # produs scalar
```

Matricile e definesc utilizand comanda **matrix**:

```
> A := matrix([[1,0],[3,2]]);
```

$$A := \begin{bmatrix} 1 & 0 \\ 3 & 2 \end{bmatrix}$$

```
> B := matrix([ [1, 0], [2, 1] ]);
```

$$B := \begin{bmatrix} 1 & 0 \\ 2 & 1 \end{bmatrix}$$

Avem urmatoarele operatii pentru matrici:

> `evalm(A + B);` # adunarea matricilor

$$\begin{bmatrix} 2 & 0 \\ 5 & 3 \end{bmatrix}$$

> `evalm(2*A);` # inmultirea cu scalar

$$\begin{bmatrix} 2 & 0 \\ 6 & 4 \end{bmatrix}$$

> `evalm(A &* B);` # inmultirea matricilor

$$\begin{bmatrix} 1 & 0 \\ 7 & 2 \end{bmatrix}$$

> `evalm(A &* (v+w));` # inmultirea unei matrici cu un vector

$$\begin{bmatrix} 2 & 6 \end{bmatrix}$$

> `det(A);` # determinant de A

$$2$$

> `evalm(A^(-1));` # calculul inversei matricii A

$$\begin{bmatrix} 1 & 0 \\ -\frac{3}{2} & \frac{1}{2} \end{bmatrix}$$

> `eigenvals(A);` # calculul valorilor proprii

$$1, 2$$

> `eigenvects(A);` # calculul vectorilor proprii

$$\left[2, 1, \left\{\begin{bmatrix} 0 & 1 \end{bmatrix}\right\}\right], \left[1, 1, \left\{\begin{bmatrix} 1 & -3 \end{bmatrix}\right\}\right]$$