LABORATOR 2: Ecuatii diferentiale

Initializare

> **restart:** sterge din memorie valori si variabile memorate anterior
> **with(DEtools):** incarca pachetul pt rezolvarea ecuatiilor diferentiale
> **with(plots):** incarca pachetul de grafica

Operatia de derivare (recapitulare)

Pentru ecuatiile diferentiale de ordin superior avem nevoie de definirea derivatelor de ordin superior. De exemplu sa consideram functia $f(x)=x^4+x^2+2$.

```
> restart:
> f:=x->x^4+x^2+2;
```

$$f:=x\to x^4+x^2+2$$

Derivata de ordinul 1 se calculeaza cu ajutorul comenzii **diff**

```
> diff(f(x),x);
```

$$4\,x^3+2\,x$$

Pentru derivatele de ordin superior se utilizeaza aceeasi comanda dar se pune variabila de mai multe ori, de exemplu pt derivata de ordinul 2 avem **diff(f(x),x,x)**

```
> diff(f(x),x,x);
```

$$12\,x^2+2$$

In cazul in care dorim calculul derivatei de ordinul 4, putem proceda ca mai inainte: **diff(f(x),x,x,x,x)** punand varibila de 4 ori sau se poate simplifica scrierea utilizand comanda: **diff(f(x),x$4)**

```
> diff(f(x),x$4);
```

$$24$$

O alta modaliate de a calcula derivata este prin utilizarea operatorului de derivare **D**

```
> D(f)(x);
```

$$4\,x^3+2\,x$$

Operatorul este utilizat atunci cind avem nevoie de valoarea derivatei intr-un anumit punct si este folosit pentru precizarea conditiilor initiale

```
> D(f)(0);
```

$$0$$

Pentru derivari de ordin superior se utilizeaza compunerea operatorului de derivare, de exemplu pentru derivata de ordinul 2 avem **(D@D)(f)(x)** sau **(D@@2)(f)(x)**. Pentru derivata de ordinul 3 avem **(D@D@D)(f)(x)** sau **(D@@3)(f)(x)**

```
> (D@D)(f)(x);
```

$$12\,x^2+2$$

```
> (D@D)(f)(2);
```

$$50$$

```
> (D@@2)(f)(x);
```

$$12\,x^2+2$$

```
> (D@@4)(f)(x);
```

Definirea si rezolvarea unei ecuatii diferentiale de ordinul 1

Fie ecuatia diferentiala de ordinul 1:

$$\frac{d}{dx} y(x) = \sin(x)\, y(x)^2 .$$

Aceasta ecuatie este o ecuatie cu variabile separabile adica este forma

$$\frac{d}{dx} y(x) = f(x)\, g(y) .$$

Ecuatia se defineste in MAPLE utilizand comanda **diff** dupa cum urmeaza:

```
> ecdif1:=diff(y(x),x) =sin(x)*(y(x))^2;
```

$$ecdif1 := \frac{d}{dx} y(x) = \sin(x)\, y(x)^2$$

Pentru a obtine solutia generala se utilizeaza comanda **dsolve(ecuatie, functie necunoscuta)**

```
> dsolve(ecdif1,y(x));
```

$$y(x) = \frac{1}{\cos(x) + \_C1}$$

Metodele incercate si utilizate de **MAPLE** pentru a obtine solutiile ecuatiilor diferentiale pot fi observate crescand **infolevel** pentru **dsolve** la 3:

```
> infolevel[dsolve]:=3;
```

$$infolevel_{dsolve} := 3$$

apoi reexecutam comanda **dsolve**:

```
> dsolve(ecdif1,y(x));
Methods for first order ODEs:
--- Trying classification methods ---
trying a quadrature
trying 1st order linear
trying Bernoulli
<- Bernoulli successful
```

$$y(x) = \frac{1}{\cos(x) + \_C1}$$

In cazul acestei ecuatii comanda dsolve a sesizat ca aceasta ecuatie este este de tip Bernoulli, adica este forma **y'(x)+P(x)*y(x)=Q(x)*(y(x))^alpha** unde **alpha** este diferit de 0 si 1. Daca dorim, putem cere in comanda **dsolve** sa se aplice metoda rezolvarii ecuatiilor separabile prin specificarea acestei optiuni dupa cum urmeaza:

```
> dsolve(ecdif1,y(x),[separable]);
Classification methods on request
Methods to be used are: [separable]
----------------------------
* Tackling ODE using method: separable
--- Trying classification methods ---
trying separable
<- separable successful
```

$$y(x) = \frac{1}{\cos(x) - \_C1}$$

Pentru a vedea care sunt metodele de rezolvare a ecuatiilor de ordinul 1 implementate in dsolve se poate da comanda:

```
> `dsolve/methods`[1];
```

$$[quadrature, linear, Bernoulli, separable, inverse\_linear, homogeneous, Chini, lin\_sym, exact, Abel, pot\_sym]$$

Pentru alte amanunte legate de comanda **dsolve** executati:

```
> ?dsolve;
```

Pentru suprimarea informatiilor suplimentare resetam **infolevel** pentru **dsolve** la **0**

```
> infolevel[dsolve]:=0;
```

$$infolevel_{dsolve} := 0$$

In unele cazuri este mai convenabil obtinerea solutiilor in forma implicita, acest lucru se poate realiza specificand in cadrul procedurii **dsolve** optiunea **implicit**. De exemplu, sa consideram ecuatia diferentiala:

$$\left(3\, y(x)^2 + e^x\right)\left(\frac{d}{dx}\, y(x)\right) + e^x\, (y(x) + 1) + \cos(x) = 0 \;:$$

```
> ecdif2:=(3*y(x)^2+exp(x))*diff(y(x),x)+exp(x)*(y(x)+1)+cos(x)  = 0;
```

$$ecdif2 := \left(3\, y(x)^2 + e^x\right)\left(\frac{d}{dx}\, y(x)\right) + e^x\, (y(x) + 1) + \cos(x) = 0$$

```
> dsolve(ecdif2,y(x),implicit);
```

$$e^x\, y(x) + e^x + \sin(x) + y(x)^3 + \_C1 = 0$$

Pentru a vedea avantajul, incercati sa rezolvati ecuatia fara a preciza optiunea **implicit**.

Reprezentarea grafica a solutiilor ecuatiilor diferentiale de ordinul 1

```
> with(plots):
```

Sa consideram, din nou, prima ecuatie:

$$\frac{d}{dx}\, y(x) = \sin(x)\, y(x)^2 \,.$$

```
> ecdif1:=diff(y(x),x)  =sin(x)*(y(x))^2;
```

$$ecdif1 := \frac{d}{dx}\, y(x) = \sin(x)\, y(x)^2$$

```
> sol1:=dsolve(ecdif1,y(x));
```

$$sol1 := y(x) = \frac{1}{\cos(x) + \_C1}$$

Rezultatul comenzii **dsolve** nu este o functie, ci este o ecuatie. Pentru manipularea solutiei exista doua alternative, fie definim functia ce reprezinta solutia (in situatia in care expresia ei nu e prea complicata), in cazul dat solutia depinde de variabila independenta *x* si constanta de integrare:

```
> y1:=(x,c)->1/(cos(x)+c);
```

$$y1 := (x, c) \to \frac{1}{\cos(x) + c}$$

sau avem acces la membrul drept utilizand comanda **rhs** (*right hand side*) si apoi comanda **unapply** pentru a construi solutia ca functie

```
> right_hand_expr:=rhs(sol1);
```

$$right\_hand\_expr := \frac{1}{\cos(x) + \_C1}$$

Prin comanda **unapply** se transforma expresia **right_hand_expr** in functie precizand variabilele acesteia:

```
> y2:=unapply(right_hand_expr,x,_C1);
```

$$y2 := (x, \_C1) \to \frac{1}{\cos(x) + \_C1}$$

Pentru reprezentarea grafica a catorva solutii dam valori constantei **c**, de exemplu **c:=2**, **c:=3** si **c:=4**

```
> plot([y1(x,2),y1(x,3),y1(x,4)],x=-2*Pi..2*Pi);
```



Daca se doreste obtinerea graficelor cu anumite culori precizate vom utiliza urmatoarea comanda:

```
> plot([y1(x,2), y1(x,3), y1(x,4)], x=-2*Pi..2*Pi, color=[black, red, blue]);
```



In cazul celei de a doua ecuatii:

$$\left(3\,y(x)^2+e^x\right)\left(\frac{d}{dx}\,y(x)\right)+e^x\,\left(y(x)+1\right)+\cos(x)=0$$

unde solutia este obtinuta in forma implicita, trebuie utilizata procedura **implicitplot**:

```
> ecdif2:=(3*y(x)^2+exp(x))*diff(y(x),x)+exp(x)*(y(x)+1)+cos(x) = 0;
```

$$ecdif2:=\left(3\,y(x)^2+e^x\right)\left(\frac{d}{dx}\,y(x)\right)+e^x\,\left(y(x)+1\right)+\cos(x)=0$$

```
> sol2:=dsolve(ecdif2,y(x),implicit);
```

$$sol2:=e^x\,y(x)+e^x+\sin(x)+y(x)^3+\_C1=0$$

Construim functia care ne da ecuatia implicita a solutiilor, in acest caz, expresia se afla in partea stanga a ecuatiei si vom folosi comanda **lhs** (letf hand side) pentru a avea acces la aceasta expresie

```
> left_hand_side:=lhs(sol2);
```

$$left\_hand\_side:=e^x\,y(x)+e^x+\sin(x)+y(x)^3+\_C1$$

De asemenea, trebuie construita expresia solutiei in forma implicita in functie de variabilele x si y. In expresia $left\_hand\_side$ trebuie inlocuit $y(x)$ cu variabila y, acest lucru se poate realiza folosind comanda **subs** astfel:

```
> lhs1:=subs( y(x)=y, left_hand_side );
```

$$lhs1 := e^x\, y + e^x + \sin(x) + y^3 + \_C1$$

```
> f:=unapply(lhs1,x,y,_C1);
```

$$f := (x, y, \_C1) \rightarrow e^x\, y + e^x + \sin(x) + y^3 + \_C1$$

```
> implicitplot([f(x,y,0)=0, f(x,y,0.5)=0, f(x,y,1)=0], x=-5..5,
y=-5..5, numpoints=10000);
```



Daca dorim reprezentarea grafica a mai multor solutii, putem genera un sir de functii *f(x,y,c)* pentru **c =-4, -19/5, ...,-1/5,   0, 1/5, 2/5, ..., 4** folosim comanda **seq** dupa cum urmeaza:

```
> sir_sol:=seq(f(x,y,i/5)=0,i=-20..20);
```

$$sir\_sol := e^x y + e^x + \sin(x) + y^3 - 4 = 0, e^x y + e^x + \sin(x) + y^3 - \frac{19}{5} = 0, e^x y + e^x + \sin(x) + y^3 - \frac{18}{5} = 0, e^x y + e^x + \sin(x) + y^3 - \frac{17}{5} = 0, e^x y + e^x + \sin(x) + y^3 - \frac{16}{5} = 0, e^x y + e^x + \sin(x) + y^3 - 3 = 0, e^x y + e^x + \sin(x) + y^3 - \frac{14}{5} = 0, e^x y + e^x + \sin(x) + y^3 - \frac{13}{5} = 0, e^x y + e^x + \sin(x) + y^3 - \frac{12}{5} = 0, e^x y + e^x + \sin(x) + y^3 - \frac{11}{5} = 0, e^x y + e^x + \sin(x) + y^3 - 2 = 0, e^x y + e^x + \sin(x) + y^3 - \frac{9}{5} = 0, e^x y + e^x + \sin(x) + y^3 - \frac{8}{5} = 0, e^x y + e^x + \sin(x) + y^3 - \frac{7}{5} = 0, e^x y + e^x + \sin(x) + y^3 - \frac{6}{5} = 0, e^x y + e^x + \sin(x) + y^3 - 1 = 0, e^x y + e^x + \sin(x) + y^3 - \frac{4}{5} = 0, e^x y + e^x + \sin(x) + y^3 - \frac{3}{5} = 0, e^x y + e^x + \sin(x) + y^3 - \frac{2}{5} = 0, e^x y + e^x + \sin(x) + y^3 - \frac{1}{5} = 0, e^x y + e^x + \sin(x) + y^3 = 0, e^x y + e^x + \sin(x) + y^3 + \frac{1}{5} = 0, e^x y + e^x + \sin(x) + y^3 + \frac{2}{5} = 0, e^x y + e^x + \sin(x) + y^3 + \frac{3}{5} = 0, e^x y + e^x + \sin(x) + y^3 + \frac{4}{5} = 0, e^x y + e^x + \sin(x) + y^3 + 1 = 0, e^x y + e^x + \sin(x) + y^3 + \frac{6}{5} = 0, e^x y + e^x + \sin(x) + y^3 + \frac{7}{5} = 0, e^x y + e^x + \sin(x) + y^3 + \frac{8}{5} = 0, e^x y + e^x + \sin(x) + y^3 + \frac{9}{5} = 0, e^x y + e^x + \sin(x) + y^3 + 2 = 0, e^x y + e^x + \sin(x) + y^3 + \frac{11}{5} = 0, e^x y + e^x + \sin(x) + y^3 + \frac{12}{5} = 0, e^x y + e^x + \sin(x) + y^3 + \frac{13}{5} = 0, e^x y + e^x + \sin(x) + y^3 + \frac{14}{5} = 0, e^x y + e^x + \sin(x) + y^3 + 3 = 0, e^x y + e^x + \sin(x) + y^3 + \frac{16}{5} = 0, e^x y + e^x + \sin(x) + y^3 + \frac{17}{5} = 0, e^x y + e^x + \sin(x) + y^3 + \frac{18}{5} = 0, e^x y + e^x + \sin(x) + y^3 + \frac{19}{5} = 0, e^x y + e^x + \sin(x) + y^3 + 4 = 0$$

```
> implicitplot([sir_sol],x=-5..5,y=-5..5,numpoints=10000);
```



Definirea si rezolvarea unei ecuatii diferentiale de ordinul 2

Pentru ecuatiile diferentiale de ordinul 2 se foloseste aceeasi metoda, se defineste ecuatia si apoi se utilizeaza comanda **dsolve**. De exemplu sa consideram ecuatia

$$\frac{d^2}{dx^2}y(x)+3\left(\frac{d}{dx}y(x)\right)+2y(x)=1+x^2$$

```
> ecdif3:=diff(y(x),x$2)+3*diff(y(x),x)+2*y(x)=1+x^2;
```

$$ecdif3:=\frac{d^2}{dx^2}y(x)+3\left(\frac{d}{dx}y(x)\right)+2y(x)=x^2+1$$

```
> dsolve(ecdif3,y(x));
```

$$y(x)=\frac{1}{2}x^2-\frac{3}{2}x+\frac{9}{4}-e^{-2x}\_C1+e^{-x}\_C2$$

Reprezentarea grafica a solutiilor ecuatiilor diferentiale de ordinul 2

```
> with(plots):
```

In cazul ecuatiei de ordinul 2,

$$\frac{d^2}{dx^2}y(x)+3\left(\frac{d}{dx}y(x)\right)+2y(x)=1+x^2,$$

reprezentarea grafica a solutiilor revine la particularizarea celor doua constante de integrare

```
> ecdif3:=diff(y(x),x$2)+3*diff(y(x),x)+2*y(x)=1+x^2;
```

$$ecdif3:=\frac{d^2}{dx^2}y(x)+3\left(\frac{d}{dx}y(x)\right)+2y(x)=x^2+1$$

```
> sol3:=dsolve(ecdif3,y(x));
```

$$sol3:=y(x)=\frac{1}{2}x^2-\frac{3}{2}x+\frac{9}{4}-e^{-2x}\_C1+e^{-x}\_C2$$

```
> right_hand_expr:=rhs(sol3);
```

$$right\_hand\_expr:=\frac{1}{2}x^2-\frac{3}{2}x+\frac{9}{4}-e^{-2x}\_C1+e^{-x}\_C2$$

```
> y1:=unapply(right_hand_expr,x,_C1,_C2);
```

$$y1:=(x,\_C1,\_C2)\rightarrow\frac{1}{2}x^2-\frac{3}{2}x+\frac{9}{4}-e^{-2x}\_C1+e^{-x}\_C2$$

```
> plot([y1(x,0,0), y1(x,-1,1), y1(x,1,-1)], x=-1..2, color=[black, blue,
```

red]);



sau putem repezenta un sir de solutii:

```
> sir_sol3:=seq(seq(y1(x,i/5,j/2),i=-2..2),j=-2..2);
```

$$
\begin{aligned}
sir\_sol3 := {} & \frac{1}{2}x^2-\frac{3}{2}x+\frac{9}{4}+\frac{2}{5}\mathrm{e}^{-2x}-\mathrm{e}^{-x},\ \frac{1}{2}x^2-\frac{3}{2}x+\frac{9}{4} \\
& +\frac{1}{5}\mathrm{e}^{-2x}-\mathrm{e}^{-x},\ \frac{1}{2}x^2-\frac{3}{2}x+\frac{9}{4}-\mathrm{e}^{-x},\ \frac{1}{2}x^2-\frac{3}{2}x+\frac{9}{4} \\
& -\frac{1}{5}\mathrm{e}^{-2x}-\mathrm{e}^{-x},\ \frac{1}{2}x^2-\frac{3}{2}x+\frac{9}{4}-\frac{2}{5}\mathrm{e}^{-2x}-\mathrm{e}^{-x},\ \frac{1}{2}x^2 \\
& -\frac{3}{2}x+\frac{9}{4}+\frac{2}{5}\mathrm{e}^{-2x}-\frac{1}{2}\mathrm{e}^{-x},\ \frac{1}{2}x^2-\frac{3}{2}x+\frac{9}{4} \\
& +\frac{1}{5}\mathrm{e}^{-2x}-\frac{1}{2}\mathrm{e}^{-x},\ \frac{1}{2}x^2-\frac{3}{2}x+\frac{9}{4}-\frac{1}{2}\mathrm{e}^{-x},\ \frac{1}{2}x^2-\frac{3}{2}x \\
& +\frac{9}{4}-\frac{1}{5}\mathrm{e}^{-2x}-\frac{1}{2}\mathrm{e}^{-x},\ \frac{1}{2}x^2-\frac{3}{2}x+\frac{9}{4}-\frac{2}{5}\mathrm{e}^{-2x} \\
& -\frac{1}{2}\mathrm{e}^{-x},\ \frac{1}{2}x^2-\frac{3}{2}x+\frac{9}{4}+\frac{2}{5}\mathrm{e}^{-2x},\ \frac{1}{2}x^2-\frac{3}{2}x+\frac{9}{4} \\
& +\frac{1}{5}\mathrm{e}^{-2x},\ \frac{1}{2}x^2-\frac{3}{2}x+\frac{9}{4},\ \frac{1}{2}x^2-\frac{3}{2}x+\frac{9}{4}-\frac{1}{5}\mathrm{e}^{-2x}, \\
& \frac{1}{2}x^2-\frac{3}{2}x+\frac{9}{4}-\frac{2}{5}\mathrm{e}^{-2x},\ \frac{1}{2}x^2-\frac{3}{2}x+\frac{9}{4}+\frac{2}{5}\mathrm{e}^{-2x} \\
& +\frac{1}{2}\mathrm{e}^{-x},\ \frac{1}{2}x^2-\frac{3}{2}x+\frac{9}{4}+\frac{1}{5}\mathrm{e}^{-2x}+\frac{1}{2}\mathrm{e}^{-x},\ \frac{1}{2}x^2-\frac{3}{2}x \\
& +\frac{9}{4}+\frac{1}{2}\mathrm{e}^{-x},\ \frac{1}{2}x^2-\frac{3}{2}x+\frac{9}{4}-\frac{1}{5}\mathrm{e}^{-2x}+\frac{1}{2}\mathrm{e}^{-x},\ \frac{1}{2}x^2 \\
& -\frac{3}{2}x+\frac{9}{4}-\frac{2}{5}\mathrm{e}^{-2x}+\frac{1}{2}\mathrm{e}^{-x},\ \frac{1}{2}x^2-\frac{3}{2}x+\frac{9}{4} \\
& +\frac{2}{5}\mathrm{e}^{-2x}+\mathrm{e}^{-x},\ \frac{1}{2}x^2-\frac{3}{2}x+\frac{9}{4}+\frac{1}{5}\mathrm{e}^{-2x}+\mathrm{e}^{-x},\ \frac{1}{2}x^2 \\
& -\frac{3}{2}x+\frac{9}{4}+\mathrm{e}^{-x},\ \frac{1}{2}x^2-\frac{3}{2}x+\frac{9}{4}-\frac{1}{5}\mathrm{e}^{-2x}+\mathrm{e}^{-x},\ \frac{1}{2}x^2 \\
& -\frac{3}{2}x+\frac{9}{4}-\frac{2}{5}\mathrm{e}^{-2x}+\mathrm{e}^{-x}
\end{aligned}
$$

```
> plot([sir_sol3],x=-1..1);
```



Rezolvarea problemelor cu valori initiale (Probleme Cauchy)

In general, rezolvarea anumitor probleme revin la determinarea unei solutii pentru o ecuatie diferentiala ce satisface anumite conditii initiale. Aceste probleme se numesc probleme cu valori initiale sau probleme Cauchy.

Problema Cauchy pentru o ecuatie diferentiala de ordinul 1 este de forma:

$$y'(x) = f(x, y(x))$$
$$y(x_0) = y_0$$

De exemplu, sa presupunem ca trebuie determinata solutia problemei Cauchy

$$\frac{d}{dx} y(x) = \sin(x)\, y(x)^2$$
$$y(0) = \frac{1}{3},$$

adica solutia reprezentata in paragraful anterior pentru constanta $c = 2$

```
> restart:with(DEtools):
> ecdif1:=diff(y(x),x) =sin(x)*(y(x))^2;
```

$$ecdif1 := \frac{d}{dx} y(x) = \sin(x)\, y(x)^2$$

Definim conditia initiala:

```
> cond_in:=y(0)=1/3;
```

$$cond\_in := y(0) = \frac{1}{3}$$

Comanda de rezolvare a problemei Cauchy este similara cu cea de rezolvare a ecuatiei la care se adauga si conditia initiala:

```
> sol1:=dsolve({ecdif1,cond_in},y(x));
```

$$sol1 := y(x) = \frac{1}{2 + \cos(x)}$$

Pentru reprezentarea grafica a solutiei se utilizeaza comanda **rhs**:

```
> y1:=unapply(rhs(sol1),x);
```

$$y1 := x \rightarrow \frac{1}{2 + \cos(x)}$$

```
> plot(y1(x),x=-2*Pi..2*Pi);
```



Se poate obtine graficul solutiei problemei Cauchy si direct utilizand comanda **DEplot**:

```
> DEplot(ecdif1,y(x),x=-2*Pi..2*Pi,[[cond_in]],stepsize=0.1);
```



Se observa ca in aceasta reprezentare apare si campul de directii impreuna cu solutia. Daca se doreste reprezentarea grafica a solutiilor pentru diverse conditii initiale, de exemplu $y(0)=\frac{1}{3}$, $y(0)=\frac{1}{4}$, $y(0)=\frac{1}{5}$, se utilizeaza aceeasi comanda specificand lista de conditii initiale:

```
> DEplot(ecdif1,y(x),x=-2*Pi..2*Pi,[[y(0)=1/3],[y(0)=1/4],[y(0)=1/5]],stepsize=0.1,linecolor=[black,red,blue]);
```



Daca dorim reprezentarea grafica doar a campului de directii se utilizeaza comanda:

```
> DEplot(ecdif1,y(x),x=-2*Pi..2*Pi,y=-1..1);
```



Acelasi rezultat se obtine utilizand si comanda **dfieldplot**:

```
> dfieldplot(ecdif1,y(x),x=-2*Pi..2*Pi,y=-1..1);
```



Problema Cauchy pentru o ecuatie diferentiala de ordinul 2 este de forma:

$$y''(x) = f(x, y(x), y'(x))$$
$$y(x_0) = a$$
$$y'(x_0) = b$$

Definirea celei de a doua conditii se face cu ajutorul operatorului de derivare **D**. De exemplu, sa determinam solutia problemei Cauchy

$$\frac{d^2}{dx^2} y(x) + 3 \left(\frac{d}{dx} y(x) \right) + 2 y(x) = 1 + x^2$$
$$y(0) = 0$$
$$y'(0) = 1$$

```
> ecdif3:=diff(y(x),x$2)+3*diff(y(x),x)+2*y(x)=1+x^2;
```

$$ecdif3 := \frac{d^2}{dx^2} y(x) + 3 \left(\frac{d}{dx} y(x) \right) + 2 y(x) = x^2 + 1$$

```
> cond_in:=y(0)=0,D(y)(0)=1;
```

$$cond\_in := y(0) = 0, D(y)(0) = 1$$

```
> sol3:=dsolve({ecdif3,cond_in},y(x));
```

$$sol3 := y(x) = \frac{1}{2} x^2 - \frac{3}{2} x + \frac{9}{4} - \frac{1}{4} e^{-2x} - 2 e^{-x}$$

Pentru reprezentarea grafica a solutiei fie utilizam metoda de constructie a solutiei cu **rhs** si **unapply** sau, direct, prin **DEplot**

```
> rhs3:=rhs(sol3);
```

$$rhs3 := \frac{1}{2} x^2 - \frac{3}{2} x + \frac{9}{4} - \frac{1}{4} e^{-2x} - 2 e^{-x}$$

```
> y3:=unapply(rhs3,x);
```

$$y3 := x \rightarrow \frac{1}{2}x^2 - \frac{3}{2}x + \frac{9}{4} - \frac{1}{4}e^{-2x} - 2e^{-x}$$

```
> plot(y3(x),x=-2..5);
```




```
> DEplot(ecdif3,y(x),x=-2..5,[[cond_in]],stepsize=0.1);
```